



CLIMATOLOGÍA Y CAMBIO CLIMÁTICO

TUTORIAL DE FERRET v6.1

1. INTRODUCCIÓN

Ferret es un software de análisis y visualización diseñado para ayudar a los oceanógrafos, meteorólogos y climatólogos en el análisis de grandes conjuntos de datos en rejilla (*gridded data*). Corre en la mayoría de los sistemas Unix, y en Windows XP/NT/9x.

Ferret fue desarrollado por el *Thermal Modeling and Analysis Project* (TMAP) y por el PMEL (*Pacific Marine Environmental Laboratory*) de la NOAA (*National Oceanic and Atmospheric Administration*), en Seattle, para analizar las salidas de sus modelos numéricos oceánicos y compararlas con los datos en rejilla observacionales. Las salidas de los modelos son generalmente de gran tamaño (varios Gigabytes) y poseen variables en 3 e incluso 4 dimensiones. Ferret además ofrece métodos matemáticos para el análisis de datos, pudiendo definir variables nuevas como función de variables del propio conjunto de datos. Además, los cálculos pueden ser aplicados sobre regiones específicas arbitrarias, y la forma de representación de gráficos es muy sencilla.

A lo largo de los años se han desarrollado una gran cantidad de paquetes informáticos para uso científico. Las características que hacen a Ferret superior a otros son su gran capacidad matemática, el procesamiento de datos en 4 dimensiones, y fundamentalmente, su fácil manejo.

Este tutorial ofrece la posibilidad de aprender algunas de las herramientas más básicas de Ferret para principiantes a través de un *tour* que se puede seguir paso a paso. Para aprender más a cerca de Ferret por cuenta propia, el Grupo de Usuarios de Ferret proporciona la posibilidad de preguntar dudas a cerca del funcionamiento del software a usuarios experimentados. Para unirse a este grupo sólo hay que enviar un e-mail a ferret-majordomo@noaa.gov e incluir un mensaje que diga simplemente (en el cuerpo del mensaje, no en el asunto) subscribe ferret_users.

La página web de Ferret es: <http://ferret.pmel.noaa.gov/Ferret>, donde se puede encontrar información on-line de todo lo referente a la instalación, documentación, tutoriales, etc. de Ferret.

2. CONCEPTOS BÁSICOS

En Ferret se considera que todas las variables están definidas sobre una rejilla. Esta rejilla le dice a Ferret cómo localizar los datos en el espacio y tiempo (o en cualquier otra dimensión que venga dada por las unidades de los ejes de la rejilla). Las variables en rejilla pueden tener desde una a cuatro dimensiones: longitud (x), latitud (y), altura (z) y tiempo (t). Para acceder a una variable Ferret debe conocer su nombre, el fichero de datos en el que está y la región de

su rejilla con la que se quiere trabajar. El comando `SET DATA (= USE)` selecciona el fichero de datos que se quiere leer. El comando `LET` permite crear variables nuevas.

Ferret es similar a Matlab, y los comandos de éste último podrían ser ejecutados en Ferret bien mediante script o secuencia de comandos (*archivo.jnl* en Ferret, *archivo.m* en Matlab), o bien mediante comandos ejecutados interactivamente después del aviso *yes?*

Suponiendo que Ferret ya está instalado en el PC, en una versión para Linux, la forma de empezar a trabajar con él es abrir una consola y teclear:

```
Ferret
```

Aparecerá:

```
yes?
```

Para ejecutar un script (*fichero.jnl*), se teclea:

```
yes? GO fichero.jnl
```

2.1 Comandos

A continuación se presenta una lista de los comandos más comunes y útiles:

<code>USE dataset/SET DATA dataset</code>	Carga los datos que van a ser analizados.
<code>SHOW DATA</code>	Muestra un resumen de las variables.
<code>SHOW GRID</code>	Examina las coordenadas de la rejilla.
<code>SET REGION</code>	Selecciona una region para ser analizada.
<code>LET</code>	Define una nueva variable.
<code>PLOT</code>	Produce una representación gráfica.
<code>CONTOUR</code>	Produce un mapa de contornos.
<code>SHADE</code>	Produce un mapa de contornos a colores.
<code>FILL</code>	Produce un mapa de contornos a colores suavizado.
<code>VECTOR</code>	Produce un mapa de vectores.
<code>GO</code>	Ejecuta comandos Ferret en un script (.jnl).
<code>STATISTICS</code>	Produce un resumen de los estadísticos de una variable.
<code>SAVE</code>	Salva los datos en formato NetCDF (.nc).
<code>LIST</code>	Muestra un listado de los datos en pantalla.
<code>!</code>	Para hacer un comentario en un fichero (.jnl).

La secuencia de operaciones en Ferret es:

1. Especificar el conjunto de datos.
2. Especificar la región.
3. Definir la variable deseada (opcional).
4. Solicitar la salida.

3. DATOS

Ferret acepta datos de entrada de ficheros ASCII y binarios y reconoce dos formatos standard de autodescripción de datos: NetCDF y TMAP. Network Common Data Format (NetCDF) es el método que se sugiere para el salvado de datos.

En este tutorial se describe cómo Ferret lee datos en formato NetCDF (*fichero.nc*) o en formato ASCII (*fichero.dat*).

3.1 DATOS NetCDF

Se va a mostrar un ejemplo usando los datos del COADS (Comprehensive Ocean /Atmosphere Data Set). Estos datos son una climatología mensual que representa un promedio de todos los datos disponibles para cada mes del año desde 1946 hasta 1989. En este ejemplo se supone que se quiere representar un contorno de la temperatura de la superficie del mar en el Océano Pacífico Ecuatorial usando esta climatología.

Primero hay que cargar el fichero de datos `coads_climatology.nc`. Si se está en el directorio donde están guardados los datos, se debe teclear:

```
yes? USE coads_climatology.nc
```

Si los datos no están en el mismo directorio donde está abierta la terminal, se debe indicar el *path* para localizar los datos en el sitio correcto, por ejemplo:

```
yes? USE "/home/usuario/Esitorio/alumnos/coads_climatology.nc"
```

o se debe cambiar al directorio donde estén los datos:

```
cd /home/usuario/Esitorio/alumnos
```

Existe una última opción que es cargar los datos directamente de internet, desde algún servidor. Por ejemplo:

```
yes? USE
"http://ferret.pmel.noaa.gov/thredds/dodsC/data/PMEL/coads_climatology.nc"
```

¿Qué variables están contenidas en el fichero de datos? ¿Cuál es la resolución de los datos? Para contestar a estas cuestiones, se utilizarán los comandos `SHOW DATA` y `SHOW GRID`.

```
yes? SHOW DATA
```

```
currently SET data sets:
l> /home/usuario/alumnos/coads_climatology.des (default)
name title I J K L
SST SEA SURFACE TEMPERATURE 1:180 1:90 1:1 1:12
AIRT AIR TEMPERATURE 1:180 1:90 1:1 1:12
SPEH SPECIFIC HUMIDITY 1:180 1:90 1:1 1:12
WSPD WIND SPEED 1:180 1:90 1:1 1:12
UWND ZONAL WIND 1:180 1:90 1:1 1:12
VWND MERIDIONAL WIND 1:180 1:90 1:1 1:12
SLP SEA LEVEL PRESSURE 1:180 1:90 1:1 1:12
```

Nota: En la segunda línea aparece un "1" que indica que el fichero especificado a continuación con su *path* completo, es el número 1 de los que están abiertos en esa sesión.

```
yes? SHOW GRID sst
```

```
GRID COADS1
name axis # pts start end
COADSX LONGITUDE 180mr 21E 19E(379)
COADSY LATITUDE 90 r 89S 89N
normal Z
TIME TIME 12mr 16-JAN 06:00 16DEC 01:20
```

Ahora se sabe que el fichero contiene 7 variables, se conoce qué variables contienen los datos y que la resolución de los datos es 2° longitud x 2° latitud incluyendo todo el globo. Ya se puede definir una región que contenga el Océano Pacífico Ecuatorial para el mes de enero, y entonces representar la SST para esa región.

Nota: Una región puede ser definida bien en términos de X, Y, Z y T, o bien en términos de los correspondientes índices I, J, K y L.

```
yes? SET REGION/X=180W:110W/Y=20S:20N/L=1
```

```
yes? CONTOUR sst
```

Alternativamente los comandos `SET REGION` y `CONTOUR` se pueden combinar como:

```
yes? CONTOUR/X=180W:110W/Y=20S:20N/L=1 sst[d=1]
!d=1 indica que la variable sst está en el fichero abierto como número 1.
```

Se obtendrá la representación gráfica mostrada en la Figura 1.




Figura 1. SEA SURFACE TEMPERATURE (Deg C)

Si en lugar de hacer la representación con contornos se desea con colores, deberá teclearse:

```
yes? FILL sst
```




Figura 2. SEA SURFACE TEMPERATURE (Deg C)

Ahora se quiere representar la presión a nivel del mar junto con el campo de vientos para el mes de julio (L=7), y para todo el globo. Para ello se utilizarán los siguientes comandos:

```
yes? CANCEL REGION      !cancela la región anteriormente seleccionada.
yes? SHADE/L=7 SLP      !representa en colores la variable llamada SLP para el mes de Julio (L=7).
yes? GO land            !representa los bordes de los continentes.
yes? VECTOR/OVER/L=7/LEN=10 UWND , VWND
                        !representa, encima del mapa de SLP, el campo de viento dado por las
                        !componentes horizontal y vertical del viento.
```




Figura 3. SEA LEVEL PRESSURE (MB)

Si se quiere salvar un fichero de datos en formato NetCDF se usará el comando SAVE. Por ejemplo, si se desea salvar sólo los datos de la sst de la región del Atlántico Norte, existen diferentes formas de hacerlo:

1ª Forma:

```
yes? SET REGION /X=90W:20E/Y=0:80N      !selecciona la región que se quiere salvar
yes? LET sst_AN=sst[L=1:12]             !define una variable nueva llamada sst_AN que corresponde a la
                                        !variable sst para la región seleccionada.
yes? SAVE /FILE=sst_atlantico_norte.nc sst_AN
                                        !salva la variable sst_AN en el fichero sst_atlantico_norte.nc
```

2ª Forma: hacerlo todo con una sola línea de comandos

```
yes? SAVE /FILE=sst_atlantico_norte.nc sst[X=90W:20E,Y=0:90N,L=1:12]
```

3.2 DATOS ASCII

El acceso a datos ASCII es por lo general bastante más complicado desde Ferret. Aquí se van a dar unas nociones básicas de cómo leer ficheros con una sola columna o varias columnas. Para acceder a ficheros de datos ASCII, se usa el comando FILE. Por ejemplo, se tiene un fichero llamado **nao.dat** que consiste en una columna que contiene la serie temporal del índice NAO de invierno en el periodo 1826-2000 (una columna con 175 valores). Si los datos tienen una columna con una sola variable se usa:

```
yes? FILE/VAR=nao nao.dat
```

con este comando, Ferret utilizará para leer los datos un eje abstracto (que suele ser el eje X con una longitud fija de 20480 puntos). Lo que interesa es que Ferret sea capaz de leer los 175 valores que tiene la serie temporal en el eje del tiempo (L). Para ello lo que se hace es definir el eje que se quiere antes de cargar los datos, y se le dice a Ferret que lea los datos pero utilizando el grid que se ha definido:

```
yes? DEFINE AXIS /T=1:175:1/UNIT=years/t0="16-JAN-1826" t nuevo
yes? DEFINE GRID /T=t nuevo g nuevo
yes? FILE /VAR=nao/GRID=g nuevo nao.dat
yes? SHOW DAT?
yes? PLOT nao
```




Figura 4. Valores de índice NAO de invierno par el periodo 1826-2000.

Si los datos contienen varias columnas con diferentes variables, simplemente se especifican los nombres de las variables:

```
yes? FILE/VAR=depth,temp,sal CTD.dat
```

Si lo que se quiere es salvar un fichero de datos en formato ASCII se usa el comando SAVE, pero especificando el formato (según los códigos de formato de FORTRAN) en el que se va a guardar la variable:

```
yes? SAVE /FILE=nao.dat/FORMAT=(8F6.3)/NOHEAD nao
```

donde el comando NOHEAD se usa para que la serie temporal se guarde sin cabeceras.

4. CÁLCULOS EN FERRET

Ya se había adelantado que en Ferret se pueden crear variables utilizando el comando LET. Además, muchas de las funciones disponibles en Fortran están también disponibles en Ferret, por ejemplo, SIN, COS, MIN, MAX, EXP, etc. La operación exponente es el símbolo ^. Por ejemplo:

```
yes? LET xsqr = x ^ 2
```

Además de las funciones disponibles, también es posible realizar diferentes transformaciones a lo largo de los ejes de una variable (por ejemplo, promedios, integraciones, suavizados, interpolaciones, etc.). Algunas transformaciones (por ejemplo el promedio "averaging") reducen el conjunto de datos a un solo valor. Cuando las transformaciones son especificadas a lo largo de más de un eje de una variable el orden de ejecución es: primero la transformación en el eje X, después Y, después Z y después T.

Algunas transformaciones válidas muy utilizadas son:

- @DIN Integral definida (suma pesada)
- @INN Integral indefinida
- @AVE Promedio
- @VAR Varianza
- @MIN Mínimo
- @MAX Máximo
- @SHF Desplazamiento (el valor por defecto es 1 punto)
- @SBX Suavizado boxcar (el valor por defecto es 3 puntos)
- @SBN Suavizado binomial (el valor por defecto es 3 puntos)
- @SUM Suma

Por ejemplo, si se quiere promediar el campo de la sst de la climatología COADS para el año entero, y representarlo:

```
yes? LET sst_promedio=sst[d=1,L=1:12@AVE]
yes? FILL sst_promedio
yes? go land
```




Figura 5. SEA SURFACE TEMPERATURE (Deg C)

5. SALVADO DE GRÁFICAS EN FERRET

Para salvar una gráfica de buena calidad en Ferret, antes de crearla se debe usar el comando SET MODE METAFILE. De esta forma Ferret creará un fichero llamado `metafile.plt` en el directorio en el que se encuentre. Si se desea que la gráfica tenga un nombre determinado sólo hay que especificarlo (`SET MODE METAFILE:grafica.plt`). Una vez que se haya salido de Ferret sólo hay que ejecutar en Linux la siguiente línea de comandos:

```
Fprint -o grafica.ps grafica.plt
```

Este comando UNIX creará un fichero postscript llamado `grafica.ps` a partir del fichero `grafica.plt`.

Además del salvado existen muchas formas de seleccionar el tamaño, posición, etc. de las gráficas. Por ejemplo, se quiere representar los datos de climatología COADS del mes de diciembre de la sst, slp, temperatura del aire y humedad específica en cuatro gráficas en una misma página. Para ello se usa el comando SET VIEW:

```
yes? USE coads_climatology.nc
yes? SHOW DATA
    currently SET data sets:
    1> /usr/local/ferret/descr/coads_climatology.des (default)
```

```

name      title      I      J      K      L
SST       SEA SURFACE TEMPERATURE  1:180  1:90  1:1  1:12
AIRT      AIR TEMPERATURE      1:180  1:90  1:1  1:12
SPEH     SPECIFIC HUMIDITY      1:180  1:90  1:1  1:12
WSPD     WIND SPEED            1:180  1:90  1:1  1:12
UWND     ZONAL WIND            1:180  1:90  1:1  1:12
VWND     MERIDIONAL WIND       1:180  1:90  1:1  1:12
SLP      SEA LEVEL PRESSURE    1:180  1:90  1:1  1:12
yes? SET MODE METAFILE:migrafica.plt
yes? SET VIEW UL      !selecciona vista arriba-izquierda (upper-left)
yes? FILL sst[d=1,l=12]
yes? GO LAND
yes? SET VIEW UR      !selecciona vista arriba-derecha (upper-right)
yes? FILL slp[d=1,l=12]
yes? GO LAND
yes? SET VIEW LL      !selecciona vista abajo-izquierda (lower-left)
yes? FILL airt[d=1,l=12]
yes? GO LAND
yes? SET VIEW LR      !selecciona vista abajo-derecha (lower-right)
yes? FILL speh[d=1,l=12]
yes? GO LAND
yes? CANCEL MODE METAFILE !guarda migrafica.plt
    
```




Figura 6.

Existen diferentes formas de especificar la posición dónde se quieren las gráficas. Un ejemplo se puede encontrar en los programas `portrait.jnl` que especifican el lugar exacto de la posición de los gráficos en una hoja de tamaño A4.

PRÁCTICA 1.

VALIDACIÓN DE UN MODELO DE CIRCULACIÓN GENERAL EN LA REGIÓN DEL ATLÁNTICO NORTE

Los modelos de circulación general del clima (GCMs) son, hasta el momento, las herramientas más avanzadas con que se cuenta en el estudio de la variabilidad climática y del cambio climático. Consisten en modelos matemáticos basados en procesos físicos que describen la interacción entre la atmósfera, los continentes y los océanos, y reproducen los diferentes procesos climáticos y de retroalimentación partiendo de un estado o condición inicial.

Para mejorar el funcionamiento de los modelos generales de circulación y probar su capacidad predictiva es imprescindible, por un lado, disponer de medidas de calidad de las variables climáticas utilizadas como *inputs*, y por otro, llevar a cabo una validación de sus resultados mediante la comparación sistemática con los datos observados. La necesidad de dicha validación ya ha sido defendida por múltiples autores (Mehel, G.A. *et al.*, 2000 y Covey, C. *et al.*, 2003; Goody *et al.*, 2002; Nieto *et al.*, 2004; Nieto and Rodríguez-Puebla, 2006; Wang *et al.*, 2006).

El objetivo de esta práctica consiste en estudiar la capacidad de un GCMs de reconocido prestigio internacional para simular el clima sobre la región del Atlántico Norte. Los resultados obtenidos proporcionarán información sobre el éxito del modelo al reproducir diferentes variables, lo que permitirá aumentar la exactitud del mismo a la hora de simular el comportamiento futuro de distintas variables climáticas en esta región. Se trata de validar las salidas del modelo, con el objetivo de acotar sus incertidumbres tanto en la reproducción del clima actual como en el futuro.

El GCM que se va a utilizar es el ECHAM5. Este modelo se desarrolló en el Instituto Max Planck de Alemania. Para encontrar una descripción detallada del modelo se puede visitar la siguiente dirección web:

http://www-pcmdi.llnl.gov/ipcc/model_documentation/ECHAM5_MPI-OM.htm.

La resolución de la parte atmosférica es de $1.9^\circ \times 1.9^\circ$ con 31 niveles en la vertical. La resolución en la parte oceánica del modelo es de $0.8^\circ \times 1.9^\circ$ y cuenta con 31 niveles en vertical.

La variable climática que se va a analizar en esta práctica, tanto para los datos procedentes de las salidas del ECHAM5 como para los datos de reanálisis es la presión a nivel del mar (SLP) en el área del Atlántico Norte [80°W - 30°E , 20°N - 80°N] durante la estación de invierno (diciembre-enero-febrero). El periodo de análisis abarca desde 1960 a 2000.

1. Comparación del campo medio de la SLP modelada con los datos de reanálisis

En esta primera etapa de la práctica se calculará el campo de valores medios y de varianzas de la SLP en invierno a partir de los datos de reanálisis del National Centers for Environmental Prediction (NCEP) y se comparará con aquel proporcionado por el modelo ECHAM5. El objetivo que se persigue al realizar esta comparación es averiguar las limitaciones y fallos del modelo al reproducir esta variable.

Se calculará el campo medio proporcionado por los datos de reanálisis, el proporcionado por el modelo y la diferencia modelo-reanálisis. En el campo de las diferencias se realizará un test-t y se señalarán aquellas zonas en las que la diferencia entre las medias es significativa al nivel de confianza del 95%. Asimismo se compararán los campos de varianza, analizando la igualdad de varianzas mediante un test F al nivel de confianza del 95%.

Lo primero que hay que hacer es recopilar los datos necesarios. Los datos de SLP del reanálisis del NCEP se pueden descargar de la página web:

<http://www.cdc.noaa.gov/data/gridded/data.ncep.reanalysis.derived.surface.html>

Se trata del fichero de datos denominado **slp.mon.mean.nc**. Se carga el fichero en Ferret y se pide que muestre el contenido del fichero cargado:

```
yes? use slp.mon.mean.nc
yes? show data
      currently SET data sets:
      l> .\slp.mon.mean.nc (default)
      name  title          I          J          K          L
      SLP   Sea Level Pressure  1:144    1:73      1:733
yes? fill slp[l=1]
yes? go land
```




Figura 7. Sea Level Pressure (millibars)

```
yes? list slp[i=144,j=73,l=1:150]
      VARIABLE : Sea Level Pressure (millibars)
      DATA SET : monthly mean slp from the NCEP Reanalysis
      FILENAME  : .\slp.mon.mean.nc
```

```
SUBSET   : 150 points (TIME)
LONGITUDE: 2.5W
LATITUDE : 90N
          2.5W
          144
01-JAN-1948 00 / 1: 1014.
01-FEB-1948 00 / 2: 1007.
01-MAR-1948 00 / 3: 1008.
01-APR-1948 00 / 4: 1015.
01-MAY-1948 00 / 5: 1016.
01-JUN-1948 00 / 6: 1015.
01-JUL-1948 00 / 7: 1015.
01-AUG-1948 00 / 8: 1009.
01-SEP-1948 00 / 9: 1009.
.....
01-MAR-1959 00 / 135: 1001.
01-APR-1959 00 / 136: 1019.
01-MAY-1959 00 / 137: 1020.
01-JUN-1959 00 / 138: 1009.
01-JUL-1959 00 / 139: 1011.
01-AUG-1959 00 / 140: 1017.
01-SEP-1959 00 / 141: 1016.
01-OCT-1959 00 / 142: 1012.
01-NOV-1959 00 / 143: 1016.
01-DEC-1959 00 / 144: 1005.
01-JAN-1960 00 / 145: 1025.
01-FEB-1960 00 / 146: 1027.
01-MAR-1960 00 / 147: 1023.
01-APR-1960 00 / 148: 1020.
01-MAY-1960 00 / 149: 1016.
01-JUN-1960 00 / 150: 1015.
```

```
yes? list slp[i=144,j=73,l=601:626]
01-JAN-1998 00 / 601: 1022.
01-FEB-1998 00 / 602: 1014.
01-MAR-1998 00 / 603: 1021.
01-APR-1998 00 / 604: 1020.
01-MAY-1998 00 / 605: 1015.
01-JUN-1998 00 / 606: 1015.
01-JUL-1998 00 / 607: 1009.
01-AUG-1998 00 / 608: 1009.
01-SEP-1998 00 / 609: 1018.
01-OCT-1998 00 / 610: 1018.
01-NOV-1998 00 / 611: 1022.
01-DEC-1998 00 / 612: 1013.
01-JAN-1999 00 / 613: 1012.
01-FEB-1999 00 / 614: 1015.
01-MAR-1999 00 / 615: 1026.
01-APR-1999 00 / 616: 1019.
01-MAY-1999 00 / 617: 1021.
01-JUN-1999 00 / 618: 1004.
01-JUL-1999 00 / 619: 1013.
01-AUG-1999 00 / 620: 1017.
01-SEP-1999 00 / 621: 1009.
01-OCT-1999 00 / 622: 1010.
01-NOV-1999 00 / 623: 1012.
01-DEC-1999 00 / 624: 1011.
01-JAN-2000 00 / 625: 1012.
01-FEB-2000 00 / 626: 1013.
```


A continuación se seleccionan los datos de la región del Atlántico Norte para el periodo deseado:

```
yes? set memory /size=200mb          !aumenta la memoria de trabajo
yes? let slp_ncep=slp[x=80w:30e,y=20n:80n,l=144:626]
yes? save /file=slp.mon.ncep_1960_2000.nc slp_ncep
yes? exit

Ferret
yes? use slp.mon.ncep_1960_2000.nc
yes? show data
      currently SET data sets:
      l> .\slp.mon.ncep_1960_2000.nc (default)
      name  title  I      J      K      L
      SLP_NCEP  SLP      1:45    1:25    ...    1:483

yes? fill slp_ncep[l=1]
yes? go land
```




Figura 8. SLP[X=80W:30E,Y=20N:80N,L=144:626]

Ya se tiene el fichero de datos con la región y periodo temporal deseados, ahora sólo hay que hacer el promedio estacional. Para ello se usará el script *estacional.jnl*

```
! PROGRAMA PARA CALCULAR LOS DATOS ESTACIONALES DE INVIERNO DE LA SLP
! DE LOS DATOS DE REANÁLISIS DEL NCEP.
!
!   go estacional.jnl
!
! Hay que tener en cuenta que se usa para invierno el diciembre del
! año anterior y que se trata de los datos mensuales desde diciembre
! de 1959 a febrero de 2000.
!
! use "/home/usuario/Esitorio/alumnos/slp.mon.ncep_1960_2000.nc"
!
show data
!
! Se aumenta la memoria de trabajo
set memory /size=200mb
!
let dic=slp_ncep[l=1:483:12]
let ene=slp_ncep[l=2:483:12]
let feb=slp_ncep[l=3:483:12]
!
!save /file=dic.nc/clobber dic
!save /file=ene.nc/clobber ene
!save /file=feb.nc/clobber feb
!
let slpinvierno=(ene[l=1:41]+feb[l=1:41]+dic[l=1:41])/3
!
save /file=slpinvierno_ncep_1960_2000.nc/clobber slpinvierno
! clobber significa que se sobrescribe en el fichero si éste ya existe
!
```

Los datos de SLP del modelo ECHAM5 ya están preparados para el invierno en el periodo 1960-2000, en el fichero denominado *slpinvierno_echam5.nc*. Lo primero que hay que hacer es cargar el fichero con Ferret y ver qué contiene exactamente este archivo:

```
yes? use slpinvierno_echam5.nc
yes? show data
      currently SET data sets:
      l> .\slpinvierno_echam5.nc (default)
      name  title  I      J      K      L
      SLPINVIERNO.....  1:60    1:33    1:1     1:41

yes? fill slpinvierno[l=1]
yes? go land
```




Figura 9. $(ENE[L=1:41]+FEB[L=1:41]+DIC[L=1:41])/3$

Se puede ver que el fichero contiene la región del Atlántico Norte deseada y que tiene la misma cobertura temporal que los datos de reanálisis, pero en este caso las unidades usadas son Pascales en lugar de mb (=hPa) y la rejilla espacial que muestran los datos del ECHAM5 ($i=1:60, j=1:33$), es diferente a la que muestran los datos del reanálisis ($i=1:45, j=1:25$). A la hora de realizar cálculos que involucren conjuntamente los dos conjuntos de datos, es necesario que ambos conjuntos de datos tengan el mismo espaciado de rejilla, ya que Ferret realiza las operaciones celdilla a celdilla. Es necesario por tanto hacer un *regrid* sobre los datos del ECHAM5 de manera que su *grid* sea el mismo que el del NCEP. Para ello vamos a usar el programa *regrid.jnl*.

```
! PROGRAMA PARA HACER UN REGRID DE LOS DATOS DEL ECHAM5 DE MANERA
! QUE SU GRID FINAL COINCIDA CON EL DE LOS DATOS DE REANÁLISIS DEL
! NCEP.
!
! go regrid.jnl
!
use "/home/usuario/Esritorio/alumnos/slpinvierno_ncep_1960_2000.nc"
use "/home/usuario/Esritorio/alumnos/slpinvierno_echam5.nc"
!
show data
!
  currently SET data sets:
!
! 1> .\slpinvierno_ncep_1960_2000.nc
! name title I J K L
! SLPINVIERNO
! (ENE[L=1:41]+..... 1:45 1:25 ... 1:41
!
```

```
! 2> .\slpinvierno_echam5.nc (default)
! name title I J K L
! SLPINVIERNO
! (ENE[L=1:41]+..... 1:60 1:33 1:1 1:41
!
set memory /size=200mb
!
define grid/like=slpinvierno[d=1] gnew
let slpinvierno_echam5rg=slpinvierno[d=2,G=gnew]
!
save/file=slpinvierno_echam5rg.nc/clobber slpinvierno_echam5rg
!
```

Para calcular los campos medios y su diferencia, así como los campos de varianzas de la SLP se usarán los programas *difmedias.jnl* y *difvarianzas.jnl*.

```
! PROGRAMA PARA HACER EL GRÁFICO DE LOS CAMPOS MEDIOS DE SLP DEL NCEP,
! DEL ECHAM5 Y LA DIFERENCIA ENTRE AMBOS
!
! go difmedias.jnl
!
use "/home/usuario/Esritorio/alumnos/slpinvierno_ncep_1960_2000.nc"
use "/home/usuario/Esritorio/alumnos/slpinvierno_echam5rg.nc"
!
show data
!
set memory /size=200mb
!
go portrait4c.jnl
!
! Se elimina el logo de Ferret en la Figura que se va a generar.
cancel mode logo
!
set mode metafile:slp_media.plt
!
! Se hace el promedio temporal de las variables en cada celdilla
let media_slp_ncep=(slpinvierno[d=1,l=@ave])
!
! Se pasan los datos del modelo a hPa dividiendo por 100 y se hace su
! promedio temporal
let slp_modelo=(slpinvierno_echam5rg[d=2])/100
let media_slp_modelo=slp_modelo[l=@ave]
!
! Se hace la diferencia entre los campos medios
let dif= media_slp_modelo-media_slp_ncep
!
save /file=dif.nc/clobber dif
!
set view v1
! Se elige el intervalo de la escala de colores de 980 hPa a 1040 hPa,
! de 2 en 2. Si salen manchas blancas en los gráficos es que no se han
! elegido correctamente los extremos de la escala.
! nolabels significa que se eliminan todas las etiquetas de
! la gráfica.
fill /nolabels/levels=(980,1040,2) media_slp_ncep
go land
!
set view v2
fill /nolabels/levels=(980,1040,2) media_slp_modelo
go land
```



```

!
! Se hace el test-t a la diferencia de medias.
! Al 95% de confianza el estadístico t es igual a 1.99 (two-tailed)
! para los grados de libertad n1+n2-2 (=80 grados de libertad)
!
let error=abs(dif/((slpinvierno[d=1,l=@var]/41)+(slp_modelo[l=@var]/41))^0.5)
!
let conf=if error gt 1.99 then 1
!
set view v3
! El intervalo elegido para la escala de colores de las diferencias va
! de -20 a 10, de 2 en 2.
fill /nolabels/levels=(-20,10,2) dif
contour /over/nolabels conf*dif
go land
!
cancel mode metafile
!

```




Figura 10. Campos medios de la SLP de invierno en el periodo 1960-2000 para los datos del: a) reanálisis NCEP y b) modelo ECHAM5. c) Diferencias entre el campo de presiones medio del NCEP y del ECHAM5. Las áreas donde las diferencias son significativas al 95% están marcadas por líneas. Unidades (hPa).

```

! PROGRAMA PARA HACER EL GRÁFICO DE LOS CAMPOS DE DESVIACIÓN TÍPICA DE
! LOS DATOS DE SLP DEL NCEP, DEL ECHAM5 Y DEL COCIENTE ENTRE VARIANZAS
!
! go difvarianzas.jnl
!
use "/home/usuario/Escritorio/alumnos/slpinvierno_ncep_1960_2000.nc"
use "/home/usuario/Escritorio/alumnos/slpinvierno_echam5rg.nc"
!
show data
!
set memory /size=200mb
!
go portrait4c.jnl
cancel mode logo
!
set mode metafile:slp_varianza.plt
!
! Se convierten los datos del modelo de Pa a hPa dividiendo por 100
let slp_modelo=(slpinvierno_echam5rg[d=2])/100
!
let desv_slp_ncep=slpinvierno[d=1,l=@var]^0.5
let desv_slp_modelo=slp_modelo[l=@var]^0.5
!
set view v1
fill /nolabels/levels=(0,7,0.5) desv_slp_ncep
go land
!
set view v2
fill /nolabels/levels=(0,7,0.5) desv_slp_modelo
go land
!
! Se calcula el cociente entre las varianzas y se aplica el test-F
! para obtener y representar las áreas donde las varianzas son
! distintas al nivel de significación del 95%:
!  $\sigma_1^2 \neq \sigma_2^2 \Rightarrow F > F_{\alpha/2, v_1, v_2}$  ó  $F < F_{1-\alpha/2, v_1, v_2}$  con  $v_1 = n_1 - 1$ ,  $v_2 = n_2 - 1$  y  $\alpha = 0.05$ 
let fttest=(slpinvierno[d=1,l=@var])/(slp_modelo[l=@var])
let conf2=if ((fttest gt 1.875) or (fttest lt 0.5333)) then 1
!
set view v3
fill /nolabels/levels=(0,5,0.5) fttest
contour /overplot/nolabels fttest*conf2
go land
!
cancel mode metafile
!

```




Figura 11. Campos de desviaciones típicas (hPa) de la SLP de invierno en el periodo 1960-2000, para los datos del: a) reanálisis del NCEP y b) modelo ECHAM5. c) Cociente entre el campo de varianzas del NCEP y del ECHAM5. Las áreas donde las varianzas son distintas al nivel de confianza del 95% están marcadas por líneas.

Las gráficas de los campos medios de la SLP (Figura 10) muestran que las diferencias significativas encontradas son fundamentalmente positivas. En el norte de África se aprecian presiones simuladas más altas que las producidas por el reanálisis. Otras diferencias positivas relevantes se sitúan sobre la mayor parte de Norte América, la Bahía de Baffin y el norte de la Península Escandinava. Sin embargo, sobre Groenlandia encontramos diferencias negativas, que indican que en esta región la presión es subestimada por el modelo.

En lo que respecta al estudio de la variabilidad, comparando las gráficas a) y b) de la Figura 11, puede observarse que, en general, el modelo subestima la variabilidad del campo de presiones y el resultado del análisis de la varianza muestra (Figura 11c) que se obtienen diferencias muy importantes que resultan significativas en el continente africano y en el Mar de Noruega.

2. Modos de variabilidad observados y modelados.

En este apartado se mostrarán los resultados obtenidos al realizar un análisis de componentes principales a los datos de presión a nivel de mar. Este análisis tiene como objetivo obtener los principales patrones de variabilidad y comprobar si el modelo reproduce los patrones hallados en los datos de reanálisis. El interés de este estudio radica en parte, en que en la obtención de proyecciones de cambio climático mediante *downscaling* estadístico, se usan como variables predictoras tales patrones de variabilidad.

Para realizar el PCA de los datos de reanálisis utilizaremos el programa `pcacorrelaciones_ncep.jnl` que se muestra a continuación.

```
! PROGRAMA PARA CALCULAR LAS COMPONENTES PRINCIPALES DE LOS DATOS DE
! SLP DEL NCEP
!
!   go pcacorrelaciones_ncep.jnl
!
! Se leen los datos del directorio en el que estén ubicados
use "/home/usuario/Escritorio/alumnos/slpinvierno_ncep_1960_2000.nc"
show data
!
set memory /size=600mb
!
! Se prepara la gráfica con 6 mapas
go portrait6d.jnl
cancel mode logo
!
! Se estandarizan los datos quitando la media y dividiendo por la
! desviación típica, así el PCA se hace sobre la matriz de
! correlaciones y no de covarianzas, que es la que usa Ferret por
! defecto.
!
let slpinvierno_normalizado=(slpinvierno[1=1:41]-
slpinvierno[1=1:41@ave])/(slpinvierno[1=1:41@var])^0.5
!
set mode metafile: eof_ncep_corr.plt
!
! Se calculan los estadísticos del PCA. El comando eof_stat calcula:
! para J=1 sale el número de EOFs (que debe ser igual al número de
! puntos),
! para J=2 calcula el porcentaje de varianza explicado por cada EOF,
! para J=3 el autovalor asociado a cada EOF.
!
let estat=eof_stat(slpinvierno_normalizado,0.3)
! El 0.3 indica que sólo se usan las series que tienen menos del 30%
! de huecos.
!
list /i=1/j=1 estat
list /i=1:6/j=2 estat
list /i=1:6/j=3 estat
!
! Se salvan los valores propios en el fichero valpropios_NCEP.dat
save /file=valpropios_NCEP.dat/format=(6F11.3)/clobber
estat[i=1:8,j=1:3]
!
! Se calculan los EOFs espaciales
let eof=eof_space(slpinvierno_normalizado,0.3)
!
! Se salvan los EOFs
save /file=eof_ncep_corr.nc/clobber eof
!
! Se representan los EOFs
!
set view v1
fill /l=1/title="EOF1"/levels=(-1,1,0.2) eof
go land
!
set view v2
fill /l=2/title="EOF2"/levels=(-1,1,0.2) eof
go land
```

```

!
set view v3
fill /l=3/title="EOF3"/levels=(-1,1,0.2) eof
go land
!
set view v4
fill /l=4/title="EOF4"/levels=(-1,1,0.2) eof
go land
!
set view v5
fill /l=5/title="EOF5"/levels=(-1,1,0.2) eof
go land
!
set view v6
fill /l=6/title="EOF6"/levels=(-1,1,0.2) eof
go land
!
cancel mode metafile
!
! Se calculan las PCs temporales (que están normalizadas)
let pc=eof_tfunc(slpinvierno_normalizado,0.3)
!
! Se salvan las PCs en un fichero .dat
save /file=pcs_ncep_corr.dat/format=(6(F8.3))/clobber pc[i=1:6]
!
! Se representan las PCs en otra ventana
!
set wind 2
set mode metafile:pcs_ncep_corr.plt
!
set view v1
plot /i=1/vlimit=-3:3:0.5/title="PC1" pc
!
set view v2
plot /i=2/vlimit=-3:3:0.5/title="PC2" pc
!
set view v3
plot /i=3/vlimit=-3:3:0.5/title="PC3" pc
!
set view v4
plot /i=4/vlimit=-3:3:0.5/title="PC4" pc
!
set view v5
plot /i=5/vlimit=-3:3:0.5/title="PC5" pc
!
set view v6
plot /i=6/vlimit=-3:3:0.5/title="PC6" pc
!
cancel mode metafile
!

```




Figura 12. EOFs de la SLP de invierno en el periodo 1960-2000 para los datos de reanálisis NCEP.

El análisis de componentes principales para los datos de SLP de reanálisis del NCEP (Figura 6) muestra un primer patrón asociado a la Oscilación del Atlántico Norte (NAO), con correlaciones de signo negativo en la parte sur de la región de estudio, más importantes sobre el Mediterráneo. El norte de la región estudiada presenta correlaciones positivas. El segundo EOF representa una estructura tripolar con un centro de correlaciones altas negativas en el Norte de Europa. El tercer EOF muestra un patrón con fuertes factores de carga positivos centrado alrededor de los 40° W y 55° N. Los patrones asociados a los siguientes EOFs muestran varias regiones con correlaciones de distinto signo.

Para calcular los EOFs de la SLP del modelo ECHAM5 se utilizará el programa *pcacorrelaciones_echam5.jnl*:

```

! PROGRAMA PARA CALCULAR LAS COMPONENTES PRINCIPALES DE LOS DATOS DE
! SLP DEL ECHAM5
!
! go pcacorrelaciones_echam5.jnl
!
! Se leen los datos en el directorio en el que estén ubicados
!
use "/home/usuario/Escritorio/alumnos/slpinvierno_echam5rg.nc"
show data
!
set memory /size=600mb
!
! Se prepara la gráfica con 6 mapas
go portrait6d.jnl
cancel mode logo
!
! Se estandarizan los datos quitando la media y dividiendo por la
! desviación típica, así el PCA se hace sobre la matriz de
! correlaciones y no de covarianzas, que es la que usa Ferret por
! defecto.
!
let slpinvierno_normalizado=(slpinvierno_echam5rg[l=1:41]-
slpinvierno_echam5rg[l=1:41@ave])/(slpinvierno_echam5rg[l=1:41@var])^0
.5
!
set mode metafile: eof_echam5_corr.plt
!
! Se calculan los estadísticos del PCA. El comando eof_stat calcula:
! para J=1 sale el numero de EOFs (que debe ser igual al número de
! puntos)
! para J=2 calcula el porcentaje de varianza explicado por cada EOF
! para J=3 el autovalor asociado a cada EOF
!
let estat=eof_stat(slpinvierno_normalizado,0.3)
! El 0.3 indica que sólo se usan las series que tienen menos del 30%
! de huecos.
!
list /i=1/j=1 estat
list /i=1:6/j=2 estat
list /i=1:6/j=3 estat
!
! Se salvan los valores propios en el fichero valpropios_ECHAM5.dat
save /file=valpropios_ECHAM5.dat/format=(6F11.3)/clobber
estat[i=1:8,j=1:3]
!
! Se calculan los EOFs espaciales
let eof=eof_space(slpinvierno_normalizado,0.3)
!
! Se salvan los EOFs
save /file=eof_echam5_corr.nc/clobber eof
!
! Se representan los EOFs
!
set view v1
fill /l=1/title="EOF1"/levels=(-1,1,0.2) eof
go land
!
set view v2

```

```

fill /l=2/title="EOF2"/levels=(-1,1,0.2) eof
go land
!
set view v3
fill /l=3/title="EOF3"/levels=(-1,1,0.2) eof
go land
!
set view v4
fill /l=4/title="EOF4"/levels=(-1,1,0.2) eof
go land
!
set view v5
fill /l=5/title="EOF5"/levels=(-1,1,0.2) eof
go land
!
set view v6
fill /l=6/title="EOF6"/levels=(-1,1,0.2) eof
go land
!
cancel mode metafile
!
! Se calculan las PCs temporales (que están normalizadas)
let pc=eof_tfunc(slpinvierno_normalizado,0.3)
!
! Se salvan las PCs en un fichero .dat
save /file=pcs_echam5_corr.dat/format=(6(F8.3))/clobber pc[i=1:6]
!
! Se representan las PCs en otra ventana
!
set wind 2
set mode metafile:pcs_echam5_corr.plt
!
set view v1
plot /i=1/vlimit=-3:3:0.5/title="PC1" pc
!
set view v2
plot /i=2/vlimit=-3:3:0.5/title="PC2" pc
!
set view v3
plot /i=3/vlimit=-3:3:0.5/title="PC3" pc
!
set view v4
plot /i=4/vlimit=-3:3:0.5/title="PC4" pc
!
set view v5
plot /i=5/vlimit=-3:3:0.5/title="PC5" pc
!
set view v6
plot /i=6/vlimit=-3:3:0.5/title="PC6" pc
!
cancel mode metafile
!

```





Figura 13. EOFs de la SLP de invierno en el periodo 1960-2000 para los datos del modelo ECHAM5.

Una forma de realizar una asignación de patrones es calcular los coeficientes de correlación entre los modos de NCEP y los del modelo. Teniendo en cuenta el número de datos de los que se dispone, el coeficiente de correlación es significativo al 95% a partir del valor 0.3. Para calcular estos coeficientes de correlación entre los EOFs obtenidos para los datos de reanálisis y para los del modelo, utilizamos el programa *regressxy.jnl*.

```
! PROGRAMA PARA CALCULAR LAS CORRELACIONES ENTRE LOS EOFs DE LOS DATOS
! DE SLP DE REANÁLISIS DEL NCEP Y LOS DEL MODELO ECHAM5
!
!   go regressxy.jnl
!
use "/home/usuario/Esitorio/alumnos/eof_ncep_corr.nc"
use "/home/usuario/Esitorio/alumnos/eof_echam5_corr.nc"
!
! Se hace la correlación en el plano xy
!
! El contador l=1:6 indica el EOF utilizado, por lo que hay que ir
! cambiando el valor de l cada vez que se ejecute el programa.
!
let p=eof[d=1,l=1]
let q=eof[d=2,l=1]
set grid q
!
! Description: define FERRET variables for regression along the X and
Y axis

say ... Linear Regression In the XY Plane
say ... Instructions:
say Use the LET command to define new variables
say Define the variable P as your independent (X) variable
say Define the variable Q as your dependent (Y) variable
say Results will be variables "SLOPE", "INTERCEP" and "RSQUARE"
say QHAT will be the regression estimate
say Note: If "Y" is your independent variable then
say ... "SET GRID Q" after defining Q.
say ...

let pq = p*q
let pqmask = pq-pq ! 0 or "missing" so all variables share the same
missing
let pmasked = p + pqmask
let qmasked = q + pqmask
let pp = pmasked*pmasked
let qq = qmasked*qmasked

let pxave = pmasked[x=@ave]
let qxave = qmasked[x=@ave]
let pave = pxave[y=@ave]
let qave = qxave[y=@ave]
let pqxave = pq[x=@ave]
let ppxave = pp[x=@ave]
let qqxave = qq[x=@ave]
let pqgave = pqxave[y=@ave]
let ppgave = ppxave[y=@ave]
let qqgave = qqxave[y=@ave]
let pvar = ppgave - pave*ppgave
let qvar = qqgave - qave*qqgave
let pqvar = pqgave - pave*qqgave

let slope = pqvar / pvar
let intercep = qave - slope*pave
let qhat = slope*p + intercep
let rsquare = (pqvar*pqvar) / (pvar*qvar)
let corr = pqvar/(pvar*qvar)^0.5

list corr
```

La siguiente tabla muestra los valores de dichas correlaciones tras ejecutar el programa regressxy.jnl para todos los EOFs del NCEP y del ECHAM5.

		ECHAM5					
		1° EOF	2° EOF	3° EOF	4° EOF	5° EOF	6° EOF
NCEP	1 ^{er} EOF	0.90	0.34	-0.10	0.06	-0.04	-0.19
	2° EOF	0.10	-0.46	0.42	0.59	0.25	0.40
	3° EOF	0.01	-0.30	-0.87	0.23	-0.06	0.09
	4° EOF	0.31	-0.64	-0.05	-0.39	0.32	0.24
	5° EOF	-0.13	-0.09	-0.23	0.32	0.46	-0.30
	6° EOF	0.01	-0.34	-0.11	-0.09	-0.48	0.55

Correlaciones espaciales de los EOFs de los datos de SLP de invierno de NCEP con los EOFs del modelo ECHAM5.

Se obtiene una correlación bastante alta entre el 1° EOF de los datos del NCEP y el 1° del modelo, y entre el 3° EOF del NCEP y el 3° del modelo. Los demás modos presentan coeficientes de correlación más bajos y no son reproducidos por el modelo de forma unívoca, sino que presentan correlaciones con varios EOFs distintos.

Ejercicio propuesto

La práctica realizada ha dado lugar a una evaluación preliminar del modelo ECHAM5 en la región del Atlántico Norte. Este primer análisis ha permitido evidenciar las limitaciones del modelo en la reproducción de la variable SLP. Sin embargo una validación adecuada necesita del estudio de más variables.

Analizar los datos de invierno de la temperatura en superficie de Europa y parte del Atlántico Norte correspondientes al reanálisis del NCEP ([tas_ncep.nc](#)) y al modelo ECHAM5 ([tas_echam5.nc](#)), para realizar una validación del modelo, similar a la llevada a cabo anteriormente con los datos de SLP, contestando a las siguientes preguntas a partir del análisis de las gráficas obtenidas:

1. ¿Estima bien el modelo el campo medio de temperaturas? ¿En qué zonas presenta peores simulaciones?
2. ¿Estima bien el modelo el campo de varianza de la temperatura en invierno? ¿Existen zonas con diferencias significativas?
3. ¿Qué zonas representan los seis primeros modos de variabilidad de las temperaturas en superficie en los datos de reanálisis? ¿Qué porcentaje de varianza explica cada uno de estos modos?
4. ¿Reproducen los datos de temperatura del modelo los principales modos de variabilidad encontrados en las observaciones? ¿Qué EOF del reanálisis correlacionan mejor (y por tanto quedan mejor simulados) con el modelo?